reset
set terminal pdfcairo font "Gill Sans,18" linewidth 12 rounded size 23in,18in

set style line 80 lt rgb "#808080"

set style line 81 lt 0
set style line 81 lt rgb "#808080"

set grid back linestyle 81
set border 3 back linestyle 80

n=100
mean = 10.0
s = 0.5
max=12.0
min=8.0
width=(max-min)/n
hist(x,width)=width*floor(x/width)+width/2.0
my_norm (x) = 88*exp(-(x-mean)*(x-mean)/((s*s)*2.0))/(sqrt(2*pi*(s*s)))

set xrange [min:max]
set yrange [0:100]
#set offset graph 0.05,0.05,0.05,0.0
set xtics min,(max-min)/5 ,max
set boxwidth width*0.9
set style fill solid 0.5
set tics out nomirror

set output "histogram.pdf"
set xlabel "Data Read From Disk (MB)"
set ylabel "Frequency"

plot "samples.csv" u (hist($1,width)):(1.0) smooth freq w boxes lc rgb"green" title "read() output", \
     my_norm(x) title "Normal Distribution" lc rgb"black"